"Wolfram Mathematica" ETP un KT uzdevumu piemēri

Dr.Sc.Ing. **Jānis Voitkāns**

Vienādojumu sistēmu risināšana

```
Solve[{I1 == Ik1, I2 == Ik1 + Ik2, I3 == Ik3}, {I1, I2, I3}]
{{I1 → Ik1, I2 → Ik1 + Ik2, I3 → Ik3}}
```

```
Dots :

R1 = 20;
R2 = 50;
R3 = 18;
R4 = 45;
R5 = 15;
R6 = 65;
R7 = 5;
R8 = 20;
E1 = 12;
E2 = 10;
E3 = 30;
J = 0.4;
```

Kirhofa vienādojumu metode

```
Iz = Solve[{I7 - I4 - I2 - I1 == 0,
    I4 + I5 - I6 == 0,
    I1 + I2 - I5 + J == 0,
    I1 (R1 + R3) - I2 R2 == E1 - E2,
    I2 R2 + I5 R5 - I4 R4 == E2,
    I4 R4 + I6 R6 + I7 (R7 + R8) == E3}, {I1, I2, I4, I5, I6, I7}] //
  Simplify
{{I1 → 0.0779295, I2 → 0.0192264, I4 → -0.0351409,
  I5 → 0.497156, I6 → 0.462015, I7 → 0.0620149}}
```

Kontūrstrāvu metode

```
Ik = Solve[{
    Ik1 (R1 + R2 + R3) - Ik2 R2 == E1 - E2,
    Ik2 (R2 + R4 + R5) - Ik1 R2 - Ik3 R4 + J R5 == E2,
    Ik3 (R4 + R6 + R7 + R8) - Ik2 R4 + J R6 == E3}, {Ik1, Ik2, Ik3}] //
  Simplify
```

```
{{Ik1 → 0.0779295, Ik2 → 0.0971559, Ik3 → 0.0620149}}
```

Mezglu potenciālu metode

$$\varphi z = \mathrm{Solve}\Big[\Big\{\varphi 1\left(\frac{1}{R1+R3}+\frac{1}{R2}+\frac{1}{R4}+\frac{1}{R7+R8}\right)-\varphi 2\left(\frac{1}{R4}\right)-\varphi 3\,\frac{1}{R2}-\varphi 3\,\frac{1}{R1+R3} == E3\left(\frac{1}{R7+R8}\right)-E1\left(\frac{1}{R1+R3}\right)-E2\left(\frac{1}{R2}\right),$$

$$\varphi 2\left(\frac{1}{R4}+\frac{1}{R5}+\frac{1}{R6}\right)-\varphi 1\,\frac{1}{R4}-\varphi 3\,\frac{1}{R5} == 0,$$

$$\varphi 3\left(\frac{1}{R1+R3}+\frac{1}{R2}+\frac{1}{R5}\right)-\varphi 1\,\frac{1}{R1+R3}-\varphi 1\,\frac{1}{R2}-\varphi 2\,\frac{1}{R5} == E1\,\frac{1}{R1+R3}+E2\,\frac{1}{R2}+J\Big\},\ \{\varphi 1,\varphi 2,\varphi 3\}\Big]\ //\ \mathrm{FullSimplify}$$

```
{{φ1 → 28.4496, φ2 → 30.031, φ3 → 37.4883}}
```

Oma likums matricu formā mezglu potenciālu metodei ($\varphi\mathrm{m}=\mathrm{Gm}^{-1}\,\mathrm{Im}$)

$$\mathrm{Gma} = \begin{pmatrix} \left(\frac{1}{R1+R3}+\frac{1}{R2}+\frac{1}{R4}+\frac{1}{R7+R8}\right) & -\frac{1}{R4} & -\left(\frac{1}{R2}+\frac{1}{R1+R3}\right) \\ -\frac{1}{R4} & \left(\frac{1}{R4}+\frac{1}{R5}+\frac{1}{R6}\right) & -\frac{1}{R5} \\ -\left(\frac{1}{R2}+\frac{1}{R1+R3}\right) & -\frac{1}{R5} & \left(\frac{1}{R1+R3}+\frac{1}{R2}+\frac{1}{R5}\right) \end{pmatrix};$$

$$\mathrm{Ima} = \begin{pmatrix} E3\,\frac{1}{R7+R8}-E1\,\frac{1}{R1+R3}-E2\,\frac{1}{R2} \\ 0 \\ E1\,\frac{1}{R1+R3}+E2\,\frac{1}{R2}+J \end{pmatrix};$$

```
φz = Inverse[Gma].Ima // N
```

```
{{28.4496}, {30.031}, {37.4883}}
```

Nelineāru vienādojumu sistēmu risināšana

21. laboratorijas darbs,
izmantojot voltampēru raksturlīknes $U = f(I)$

$36V$ lampai

Ar interpolācijas palīdzību mērījumu voltampēru raksturlīknes tabula attiecīgajai lampai tiek pārvērsta par nepārtrauktu analītisku funkciju

```
U1 = Interpolation[{{0, 0}, {0.65, 1.72}, {0.86, 3.54},
   {1.02, 5.45}, {1.16, 7.12}, {1.31, 9.05}, {1.44, 11.03},
   {1.56, 12.75}}]
```

```
InterpolatingFunction[ Domain: {{0., 1.56}}
                       Output: scalar ]
```

Tiek pārbaudīts, kāda izskatās voltampēru raksturlīknes analītiskā funkcija

```
Plot[U1[Iz], {Iz, 0, 1.56}, AxesLabel → {"I(A)", "U(V)"},
 GridLines -> Automatic]
```



$127V$ lampai

```
U2 = Interpolation[{{0, 0}, {0.05, 15.24}, {0.08, 30.28},
   {0.1, 44.5}, {0.12, 60.62}, {0.14, 75.3}, {0.15, 90}}]
```

```
InterpolatingFunction[ Domain: {{0., 0.15}}
                       Output: scalar ]
```

```
Plot[U2[Iz], {Iz, 0, 0.15}, AxesLabel → {"I(A)", "U(V)"},
 GridLines -> Automatic]
```



120V lampai

```
U3 = Interpolation[{{0, 0}, {0.06, 15.07}, {0.15, 29.7},
   {0.26, 45.67}, {0.37, 60.3}, {0.5, 75.8}, {0.62, 90}}]
```

InterpolatingFunction[Domain: {{0., 0.62}} Output: scalar]

```
Plot[U3[Iz], {Iz, 0, 0.62}, AxesLabel → {"I(A)", "U(V)"},
 GridLines -> Automatic]
```



Tiek atrisināta nelineāru vienādojumu sistēma attiecībā pret ieejas spriegumu E1=50V, un tiek iegūtas visas ķēdes strāvas šim darba režīmam.

```
E1 = 50;
FindRoot[{I1 == I2 + I3,
  U1[I1] + U2[I2] == E1,
  U1[I1] + U3[I3] == E1}, {I1, 1}, {I2, 0.1}, {I3, 0.1}]
```

```
{I1 → 0.394748, I2 → 0.106356, I3 → 0.288392}
```

Nomērītā strāva I3 = 0.28 A

21. laboratorijas darbs, izmantojot pretestību raksturlīknes

```
R1 = Interpolation[{{0, 0.5}, {0.65, 1.72 / 0.65},
   {0.86, 3.54 / 0.86}, {1.02, 5.45 / 1.02}, {1.16, 7.12 / 1.16},
   {1.31, 9.05 / 1.31}, {1.44, 11.03 / 1.44}, {1.56, 12.75 / 1.56}}]
```

```
InterpolatingFunction[ Domain: {{0., 1.56}}
                       Output: scalar ]
```

```
Plot[R1[x], {x, 0, 1.56}, PlotRange → All, AxesOrigin → {0, 0},
 AxesLabel → {"I(A)", "R(Ω)"}, GridLines -> Automatic]
```



```
R2 = Interpolation[{{0, 200}, {0.05, 15.24 / 0.05},
   {0.08, 30.28 / 0.08}, {0.1, 44.5 / 0.1}, {0.12, 60.62 / 0.12},
   {0.14, 75.3 / 0.14}, {0.15, 90 / 0.15}}]
```

```
InterpolatingFunction[ Domain: {{0., 0.15}}
                       Output: scalar ]
```

```
Plot[R2[x], {x, 0, 0.15}, PlotRange → All, AxesOrigin → {0, 0},
 AxesLabel → {"I (A)", "R (Ω)"}, GridLines -> Automatic]
```




```
R3 = Interpolation[{{0, 280}, {0.06, 15.07 / 0.06},
   {0.15, 29.7 / 0.15}, {0.26, 45.67 / 0.26}, {0.37, 60.3 / 0.37},
   {0.5, 75.8 / 0.5}, {0.62, 90 / 0.62}}]
```

```
InterpolatingFunction[ Domain: {{0., 0.62}}
                       Output: scalar ]
```

```
Plot[R3[x], {x, 0, 0.62}, PlotRange → All, AxesOrigin → {0, 0},
 AxesLabel → {"I (A)", "R (Ω)"}, GridLines -> Automatic]
```



```
FindRoot[{I1 == I2 + I3,
  I1 R1[I1] + I2 R2[I2] == E1,
  I1 R1[I1] + I3 R3[I3] == E1}, {I1, 1}, {I2, 0.1}, {I3, 0.1}]
```

```
{I1 → 0.394581, I2 → 0.10631, I3 → 0.288271}
```

```
Nomērītā strāva  I3 = 0.28 A
```

Magnētisko ķēžu mājdarbs

Tiek interpolēta galvenās magnetizācijas līknes tabula, simetriski parādot arī negatīvo zaru. Rezultātā tiek iegūta analītiska, nepārtraukta galvenā magnetizācijas līkne

```
B = Interpolation[{{-1600, -1.60}, {-800, -1.57}, {-600, -1.53},
   {-400, -1.47}, {-200, -1.28}, {-120, -1.14}, {-80, -1.02},
   {-60, -0.93}, {-40, -0.75}, {-20, -0.22}, {0, 0},
   {20, 0.22}, {40, 0.75}, {60, 0.93}, {80, 1.02}, {120, 1.14},
   {200, 1.28}, {400, 1.47}, {600, 1.53}, {800, 1.57},
   {1600, 1.60}}, InterpolationOrder → 2]
```

```
InterpolatingFunction[ Domain: {{-1.60×10^3, 1.60×10^3}}
                       Output: scalar ]
```

Pārbaudam, kāda tā izskatās

```
Plot[B[H], {H, -1600., 1600.}, AxesLabel → {"H(A/m)", "B(T)"}]
```



Uzzīmējam ekvivalento magnētiskās ķēdes shēmu un sastādām shēmas vienādojumu sistēmu. Uzdodam datus un aprēķinām iegūto nelineāro vienādojumu sistēmu, magnētisko indukciju B ievadot miliTeslās. Ja ir papildnosacījums, tad to arī pievienojam vienādojumu sistēmai.

```
l1 = 0.4;
l2 = 0.2;
l3 = 0.4;
lg = 1.×10^-3;
w1 = 600;
w3 = 800;
S1 = 15.×10^-4;
S2 = 10.×10^-4;
S3 = 15.×10^-4;
I1 = 0.5;
I3 = 0.25;
μo = 4 π 10^-7;
Clear[Uab, ϕ1, ϕ2, ϕ3]
A1 = FindRoot[{ϕ1 + ϕ3 == ϕ2,
    ϕ1 == B[H1] S1 10^3,
    ϕ2 == B[H2] S2 10^3,
    ϕ3 == B[H3] S3 10^3,
    Uab == I1 w1 - H1 l1 - B[H1] lg/μo,
    Uab == H2 l2,
    Uab == I3 w3 - H3 l3},
   {{Uab, 10}, {H1, 10}, {H2, 10}, {H3, 10}, {ϕ1, 1}, {ϕ2, 1},
    {ϕ3, 1}}]
Hg = B[H1]/μo /. A1

{Uab → 177.185, H1 → 15.6939, H2 → 885.925,
 H3 → 57.0376, ϕ1 → 0.219668, ϕ2 → 1.58319, ϕ3 → 1.36352}

116537.
```

Diferenciālvienādojumu sistēmas atrisināšana, pārejas procesiem, ko izraisa komutācija

Klasiskā metode

Diferenciālvienādojumi jārisina absolūtās precizitātes režīmā. Dotie lielumi jāuzdod kā veseli skaitļi ar pakāpes reizinātāju. Nevienu skaitli nedrīkst uzdot kā decimālskaitli. Vienādojumu sistēmā jāuzdod shēmas vienādojumi un sākuma nosacījumi spriegumam Uc[0] un strāvai I_L[0], kas pakļaujas komutācijas likumiem. Jāparāda, ka Ic=C $\frac{dUc}{dt}$ un U_L=L $\frac{dI_L}{dt}$. Rezultāts beigās jāpārvērš par decimāldaļskaitli ar //N//Expand//Simplify.

```
E1 = 150;
L = 4 × 10^-3;
C1 = 5 × 10^-6;
R1 = 10;
R2 = 20;
R3 = 40;
R4 = 10;

a1 = DSolve[{i4[t] == i1[t] + i5[t],
      i4[t] == i2[t] + i3[t],
      i4[t] (R1 + R4) + uc[t] + UL[t] == E1,
      i5[t] (R2) - UL[t] == 0,
      uc[t] - i3[t] R3 == 0,

      i2[t] == C1 uc'[t],
      UL[t] == L i1'[t],

      uc[0] == 150, i1[0] == 0},
     {i1[t], i2[t], i3[t], i4[t], i5[t], uc[t], UL[t]}, t] // N //
   Expand // Simplify
```

$$\{\{i1[t] \to 2.5 + 5.\ e^{-7500.\,t} - 7.5\ e^{-5000.\,t},$$
$$i2[t] \to -7.5\ e^{-7500.\,t} + 3.75\ e^{-5000.\,t},$$
$$i3[t] \to 2.5 + 5.\ e^{-7500.\,t} - 3.75\ e^{-5000.\,t},$$
$$i4[t] \to 2.5 - 2.5\ e^{-7500.\,t},\ i5[t] \to -7.5\ e^{-7500.\,t} + 7.5\ e^{-5000.\,t},$$
$$uc[t] \to 100. + 200.\ e^{-7500.\,t} - 150.\ e^{-5000.\,t},$$
$$UL[t] \to -150.\ e^{-7500.\,t} + 150.\ e^{-5000.\,t}\}\}$$

Uzzīmējam iegūtās funkcijas

```
Plot[{i1[t] /. a1, i4[t] /. a1, i5[t] /. a1}, {t, 0, 0.001},
 PlotRange → All, GridLines → Automatic,
 PlotStyle → {{Black, Thin}, {Blue, Thin}, {Red, Thin}},
 GridLinesStyle → Directive[Black, Dotted, Thin],
 AxesLabel → {Text["t"], Text["i1(A)   i4(A)   i5(A)"]},
 PlotLegends → {"i1", "i4", "i5"}]
```



Operatoru metode pārejas procesa aprēķinam

Aprēķinu jāveic absolūtās precizitātes režīmā

Dots:

In[1]:=
```
E1 = 150;
L = 4 × 10^-3;
C1 = 5 × 10^-6;
R1 = 10;
R2 = 20;
R3 = 40;
R4 = 10;
uc0 = 75;
i0 = 0;
```

Aprēķina shēmas strāvu polinomus

In[10]:=
$$Ik = \text{Solve}\Big[\Big\{Ik1\,(R1+R4+R2+R3) - Ik2\,R2 + Ik3\,(R1+R2+R4) == \frac{E1}{p},$$
$$Ik2\,(R2+p\,L) - Ik1\,R2 - Ik3\,R2 == L\,i0,$$
$$Ik3\left(R1+R2+R4+\frac{1}{p\,C1}\right) - Ik2\,R2 + Ik1\,(R1+R2+R4) == \frac{E1}{p} - \frac{uc0}{p}\Big\},$$
$$\{Ik1, Ik2, Ik3\}\Big] \text{ // Simplify}$$
$$I4 = Ik1 + Ik3 \text{ /. } Ik \text{ // Simplify}$$

Out[10]=
$$\Big\{\Big\{Ik1 \to \frac{15\,(50\,000\,000 + 12\,500\,p + p^2)}{8\,p\,(37\,500\,000 + 12\,500\,p + p^2)},$$
$$Ik2 \to \frac{9375\,(10\,000 + p)}{p\,(37\,500\,000 + 12\,500\,p + p^2)},$$
$$Ik3 \to \frac{9375}{2\,(37\,500\,000 + 12\,500\,p + p^2)}\Big\}\Big\}$$

Out[11]=
$$\Big\{\frac{15\,(10\,000 + p)}{8\,p\,(7500 + p)}\Big\}$$

Veic inverso Laplasa transformāciju

In[12]:=
```
It =
 InverseLaplaceTransform[{Ik1 /. Ik, Ik2 /. Ik, Ik3 /. Ik, I4},
    p, t] // Expand // N // Simplify
```

Out[12]=

$$\{\{2.5 + 1.25\, e^{-7500.\, t} - 1.875\, e^{-5000.\, t}\},\ \{2.5 + 1.25\, e^{-7500.\, t} - 3.75\, e^{-5000.\, t}\},\ \{-1.875\, e^{-7500.\, t} + 1.875\, e^{-5000.\, t}\},\ \{2.5 - 0.625\, e^{-7500.\, t}\}\}$$

Vizualizē pārejas procesu grafikus

In[13]:=
```
Plot[{It[[1]], It[[2]], It[[3]], It[[4]]}, {t, 0, 0.0008},
 PlotRange → All, AxesLabel → {"t(s)", "I(A)"},
 PlotStyle → {{Black, Thin}, {Blue, Thin}, {Red, Thin},
   {Green, Thin}}, GridLines → Automatic,
 PlotLegends → {"ik1", "ik2", "ik3", "i4"}]
```

Out[13]=



Furjē transformācija, *29. lab.* darbs

Ir veikti mērījumi, atrasta pārvades funkcija H(*ω*), aprēķināts U2'(*ω*) tabulas veidā.

U2'(*ω*) tiek interpolēts, lai no tabulas iegūtu nepārtrauktu funkciju. Tabulā tiek pievienota ailīte *ω*=0, kurai spriegumu piemeklē, lai sprieguma lēciens pie t=0 būtu aptuveni 1, un viena ailīte pie *ω*=10000, kur pieņemam, ka spriegums U2'(*ω*) ir sasniedzis 0:

```
U2 = Interpolation[{{0, 0.0048228}, {125, 0.00196},
   {150, 0.00181}, {200, 0.00167}, {250, 0.0015}, {300, 0.00132},
   {400, 0.00105}, {500, 0.00084}, {750, 0.00051},
   {1000, 0.00031}, {1250, 0.00022},
   {1750, 0.00013}, {2000, 0.00009}, {2500, 0.00007},
   {3000, 0.00004}, {10000, 0}}, InterpolationOrder → 1]
```

InterpolatingFunction[Domain: $\{\{0., 1.00 \times 10^4\}\}$ Output: scalar]

Tiek pārbaudīta iegūtā nepārtrauktā funkcija

```
Plot[U2[ω], {ω, 0, 10000}, PlotRange → Full]
```



Tiek aprēķināts apgrieztās Furjē transformācijas integrālis reālam argumentam (dots Dūmiņa grāmatā). Ar //N jāparāda, ka integrālis jārēķina skaitliski. Rezultātā tiek iegūta tabula {t, U2(t)}, kas nosaukta par TabU2.

```
TabU2 = {{0, 0}};
Do[{U2t = 2/π ∫_0^10000 U2[ω] Cos[ω t] dω // N, Print[U2t],
  tU2 = {t, U2t}, TabU2 = Append[TabU2, tU2]}, {t, 0, 0.03, 0.002}]
```

```
1.
0.46368
0.258961
0.165178
0.134361
0.108763
0.0920192
0.0809077
0.0722356
0.065106
0.0589696
0.0521614
0.0432662
0.0365473
0.0320693
0.0262054
```

Uzzīmējam iegūto laika fumkciju

```
ListLinePlot[TabU2, PlotRange → Full, GridLines → Automatic,
 GridLinesStyle → Directive[Gray, Dashed],
 AxesLabel → {"t(s)", "U2(V)"}]
```



Ir iegūta komutācijas procesa laika funkcija. Var arī rēķināt katru integrāli atsevišķi savam laika momentam, bez cikla operatora Do[ ].

Ja nepieciešams tikai grafiks, bet nav vajadzīgas konkrētās sksitliskās vērtības, tad var darīt sekojoši:

$$U2t = \frac{2}{\pi}\int_0^{10000} U2[\omega]\, Cos[\omega\, t]\, d\omega;$$

```
Plot[U2t, {t, 0, 0.03}, PlotRange → Full, AxesOrigin → {0, 0},
 GridLines → Automatic, GridLinesStyle → Directive[Gray, Dashed],
 AxesLabel → {"t(s)", "U2(V)"}]
```



Redzams, ka grafiks veidojas plūstošāks, kvalitatīvāks, kā iepriekšējā gadījumā.

Nesinusoidāla periodiska signāla Furjē spektrs *(*Laboratorijas darbs Nr. *15)*

Tiek veikti mērījumi.

1. Ir iegūtas nesinusoidālā sprieguma līkņu fotogrāfijas

a. pamatsignālam,

b. taisngrieztajam signālam,

c. ar virknes kontūru filtrētajam signālam,

d. ar paralēlo kontūru filtrētajam signālam.

2. No fotogrāfijām tiek uzņemtas sprieguma U(V) un laika t(ms) koordinātes viena perioda laikā un izveidotas U=f(t) tabulas katram signālam. Izmantojot komandu Interpolation[], tabulētās līknes tiek pārvērstas par analītiskām līknēm.

Piemēram: Iegūsim spektra Furjē komponentes ar paralēlo kontūru filtrētajam signālam (d. mērījums).

No fotogrāfijas nolasītās U=f(t) tabula ir sekojoša (katrā figūriekavā pirmais skaitlis ir arguments, otrais skaitlis ir funkcijas vērtība):

```
Udt={{0,0},{1.19,10},{2.99,2},{3.28,0},{4.48,-13},{5.67,-19.5},{6.87,-13.5},{7.46,0},{8.06,20},{8.66,28},{9.4,20},{10.4,0},{11,-9},{11.9,-12},{13.1,-7},{13.7,-0},{14.6,9},{16.1,17},{17,11},{17.5,0},{17.9,-18},{18.5,-29},{19.4,-18},{20,0}};
```

Sastādot tabulu ir jāparāda perioda sākuma un beigu punkts ({0, 0} un {20, 0}), katrs punkts, kur līkne šķērso laika asi, jāparāda katrs pozitīvais vai negatīvais funkcijas maksimums ar vismaz vienu raksturīgu punktu pirms un pēc maksimuma. Vēlams tabulā iekļaut arī funkcijas pārliekuma punktus.

Tabulētā līkne tiek pārvērsta par nepārtrauktu analītisku līkni un nosaukta par Ud.

```
Ud = Interpolation[Udt]

InterpolatingFunction[ Domain: {{0., 20.}}
                       Output: scalar ]
```

Uzzīmējam interpolēto līkni un pārbaudām, vai atbilst fotogrāfijai

```
Plot[Ud[t], {t, 0, 20}]
```



3. Aprēķinām harmoniku sinusa komponenšu, kosinusa komponenšu, hatmoniku amplitūdu un sākuma leņķu lielumus. Uzdodam perioda garumu, nepieciešamo harmoniku nummurus un pirmās harmonikas leņķisko frekvenci.

$T = 20;$

$k = \{1, 2, 3, 4, 5, 6, 7, 8, 9\};$

$\omega = \frac{2\pi}{T};$

$Udo = \frac{1}{T} \int_0^T Ud[t]\, dt \; // \; N$

$A_k = \frac{2}{T} \int_0^T Ud[t]\, Sin[k\, \omega\, t]\, dt \; // \; N$

$B_k = \frac{2}{T} \int_0^T Ud[t]\, Cos[k\, \omega\, t]\, dt \; // \; N$

$Ud_k = \sqrt{A_k{}^2 + B_k{}^2}$

$\psi_k = ArcTan\left[\frac{B_k}{A_k}\right]$

```
-0.814676

{-2.87408, -0.716699, 16.5703, 0.406049,
 5.42205, 0.477835, 0.2014, -0.0486127, -0.817565}

{-2.8701, 1.77237, -7.56905, 0.91467,
 3.26789, 0.79083, 2.57579, 0.423091, 0.143449}

{4.06175, 1.91179, 18.2171, 1.00075,
 6.3307, 0.92398, 2.58366, 0.425875, 0.830054}

{0.784706, -1.18653, -0.428482, 1.153,
 0.542406, 1.02728, 1.49277, -1.4564, -0.17369}
```

4. Sasummējam harmonikas un pārbaudām, vai var iegūt sākuma līkni

Leņķi ir jākoriģē atkarībā no A_k zīmes. Ja A_k ir negatīvs, tad leņķim ψ_k jāpieskaita 180 °

$U = Udo + Ud_k[[1]]\, Sin[\omega\, t + \psi_k[[1]] + 180\,°] +$
$Ud_k[[2]]\, Sin[2\,\omega\, t + \psi_k[[2]] + 180\,°] + Ud_k[[3]]\, Sin[3\,\omega\, t + \psi_k[[3]]] +$
$Ud_k[[4]]\, Sin[4\,\omega\, t + \psi_k[[4]]] + Ud_k[[5]]\, Sin[5\,\omega\, t + \psi_k[[5]]] +$
$Ud_k[[6]]\, Sin[6\,\omega\, t + \psi_k[[6]]] + Ud_k[[7]]\, Sin[7\,\omega\, t + \psi_k[[7]]] +$
$Ud_k[[8]]\, Sin[8\,\omega\, t + \psi_k[[8]] + 180\,°] +$
$Ud_k[[9]]\, Sin[9\,\omega\, t + \psi_k[[9]] + 180\,°];$

```
Plot[U, {t, 0, 20}]
```



Redzams, ka pirmo deviņu harmoniku summārā līkne ir ļoti tuva sākuma līknei.